

第3章

演算と編集

1 算術演算

例題2 都道府県ごとの人口密度

県ファイルを読んで、各都道府県ごとの人口密度を求める一覧表を作成しなさい。

▼入力形式

県コード	県名	面積	人口
数字2桁	英数字8桁	数字5桁	数字5桁

▼処理条件

(1) 人口密度は下記の式で計算する。

$$\text{人口密度} = \text{人口} * 1000 / \text{面積}$$

(2) 面積・人口・人口密度は、出力結果のように、有効数字の前の0は空白におきかえ、3桁ごとにコンマを打つ。

▼出力形式

	0	1	2	3	4	5
	1234567890	1234567890	1234567890	1234567890	1234567890	1234567890
1						
2		(県コード)	(県名)	(面積)	(人口)	(人口密度)
3		99	XXXXXXXX	ZZ,ZZ9	ZZ,ZZ9	ZZ,ZZ9
4		99	XXXXXXXX	ZZ,ZZ9	ZZ,ZZ9	ZZ,ZZ9
5		}	}	}	}	}
6						


注 ZについてはP.28で説明する。

▼出力結果

01	ホツカイトウ	83,520	5,646	67
02	アオモリ	9,619	1,529	158
03	イワテ	15,277	1,433	93
}	}	}	}	}


●問題の分析と考え方●

この例題では、入力データの「面積」と「人口」を用いて「人口密度」を計算しなければならない。また、出力結果のように、不要な0は空白に置きかえ、3桁ごとにコンマを打って出力する。



記憶装置

●流れ図とプログラム●



```

IDENTIFICATION          DIVISION.
PROGRAM-ID.              REI0301.

* ENVIRONMENT            DIVISION.
INPUT-OUTPUT             SECTION.
FILE-CONTROL.
    SELECT KEN-FILE      ASSIGN D1.
    SELECT ITIRAN-FILE   ASSIGN SPR.

* DATA                  DIVISION.
FILE                     SECTION.
FD KEN-FILE.
01 KEN-REC.
    05 KENCODE          PIC      9(02).
    05 KENMEI           PIC      X(08).
    05 MENSEKI          PIC      9(05).
    05 JINKO            PIC      9(05).
FD ITIRAN-FILE.
01 ITIRAN-REC          PICTURE X(132).
WORKING-STORAGE SECTION.
01 E-FLG               PIC      X(03).
01 MEISAI-GYO.
    05                  PIC      X(09)  VALUE SPACE.
    05 M-KENCODE        PIC      9(02).
    05                  PIC      X(05)  VALUE SPACE.
    05 M-KENMEI         PIC      X(08).
    05                  PIC      X(05)  VALUE SPACE.
    05 M-MENSEKI        PIC      ZZ,ZZ9.
    05                  PIC      X(05)  VALUE SPACE.
    05 M-JINKO          PIC      ZZ,ZZ9.
    05                  PIC      X(05)  VALUE SPACE.
    05 M-MITUDO         PIC      ZZ,ZZ9.

* PROCEDURE              DIVISION.
SYORI.
    OPEN      INPUT      KEN-FILE      OUTPUT ITIRAN-FILE
    MOVE      "OFF"      TO      E-FLG
    PERFORM   UNTIL      E-FLG = "ON"
        READ KEN-FILE
        AT END
            MOVE      "ON"      TO      E-FLG
        NOT AT END
            MOVE      KENCODE    TO      M-KENCODE
            MOVE      KENMEI     TO      M-KENMEI
            MOVE      MENSEKI    TO      M-MENSEKI
            MOVE      JINKO      TO      M-JINKO
            COMPUTE M-MITUDO = JINKO * 1000 / MENSEKI
            WRITE ITIRAN-REC FROM MEISAI-GYO AFTER 1
    END-READ
END-PERFORM
CLOSE      KEN-FILE      ITIRAN-FILE
STOP      RUN.
  
```


●プログラムの解説●

編集

印字された数字を見やすくするために、有効数字の前の0を空白に置きかえたり、3桁ごとにコンマ「,」を入れたりすることを編集という。編集は、作業場所節の明細行で、PICTURE句において編集内容を指定しておき、入力領域から作業領域へその内容が転記されたときに、受取り側で編集される。

編集を行うための文字を、編集用文字という。編集可能な文字列の長さは30字までであり、また数字の入る桁数は18桁までである。

ここでは、ゼロ抑制のための「Z」と、数字を3桁ごとに区切るコンマ「,」を学習する。



送出し側と受取り側の関係を、いろいろな例をとおして示す。

MOVE A TO B

送出し側(A)		受取り側(B)		備 考
PICTURE 句	記憶内容	PICTURE 句	転記の結果	
9 (06)	001234	9 (06)	001234	何も編集されず、数字がそのまま 有効数字の前の0が空白に 1と2の間に「,」が入る 「,」も空白に変わる 右側から入り、先頭の2桁が切り捨てられる 最後の0だけ残る すべて空白に変わる
9 (06)	001234	ZZZZZ 9	△△1234	
9 (06)	001234	ZZZ, ZZ 9	△△1, 234	
9 (06)	000123	ZZZ, ZZ 9	△△△△123	
9 (08)	12345678	ZZZ, ZZ 9	345, 678	
9 (04)	0000	Z, ZZ 9	△△△△0	
9 (04)	0000	Z, ZZZ	△△△△△	

▲編集用文字「Z」と「,」を使った編集例

COMPUTE 文

COMPUTE 文

COMPUTE データ名 = 算術式


四則演算を行う文である。等号の左辺は算術演算を行った結果が入る領域を示す。等号の右辺は算術式といい、いろいろな計算式を記述する。

算術式で用いる演算のための記号を算術演算子といい、数学の代数式と同じように、演算順序が決まっている。
等号や算術演算子の前後には必ず1桁以上の空白を取る。
また、演算順序の変更や、算術式を見やすくするためには()を用いる。

演算の優先順位が同じ場合は、左から右へ順番に演算される。複数の()に囲まれている場合は、中の()から演算を行う。


算術演算子	意 味	優先順位
**	べき乗	1
*	乗算	2
/	除算	2
+	加算	3
-	減算	3

▲算術演算子



算術式は、データ名や数字定数を算術演算子でつないだものである。たとえば、次に示す等号の右辺はいずれも算術式である。

- ① COMPUTE A = B
- ② COMPUTE C = 3.14
- ③ COMPUTE D = EE * FF / GG
- ④ COMPUTE H = H + J



等号の右辺と左辺の関係は、等号の右辺で計算した結果を、左辺に代入することを意味する。したがって、①や②のように、データ名や数字定数が1つだけでもかまわないし、④のように左辺と右辺に同じデータ名があってもよい。

①と④の具体的な動きは次のようである。

- ① A 10 B 10 Bの内容をAに代入する。
その結果AもBも10になる。

したがって、
COMPUTE A = B と
MOVE B TO A

は同じことを意味する。

- ④ H 5 (演算前) J 1 H + J
H 6 (演算後) ← (5 + 1)

算術式の中に書かれるデータ名は数字項目でなければならない。

例 COMPUTE CC = AA * BB

正しい { AA PIC 9 (03). 誤り { AA PIC ZZZ9.
 { BB PIC 9 (05). { BB PIC 9 (03).


等号の左辺のデータ名は、編集された項目でもかまわない。

例題2のプログラムにおいて、人口密度の計算はこれを利用している。

例 COMPUTE M-MITUDO = JINKO * 1000 / MENSEKI

↑編集された項目
 ↑数字項目
 ↑数値定数
 ↑数字項目

WRITE 文



作業領域に記憶されている内容を、出力領域に転記し、出力する文である。

例題1ではWRITE文の前にMOVE文を用いて、作業領域の内容を出力領域に転記したが、FROM句を用いることにより、転記と出力を1つの文で行うことができる。

例 MOVE MEISAI-GYO TO ITIRAN-REC

WRITE ITIRAN-REC AFTER 1

↓ 1つの文にする

WRITE ITIRAN-REC FROM MEISAI-GYO AFTER 1

●練習問題●

1 次のMOVE文実行後の受取り側の内容はどうなっているか答えなさい。

MOVE A TO B

送出し側(A)		受取り側(B)	
PICTURE 句	記憶内容	PICTURE 句	転記後の内容
PIC X (05)	ABCDE	PIC X (03)	ア
PIC X (05)	ABCDE	PIC X (07)	イ
PIC X (05)	00123	PIC X (06)	ウ
PIC X (05)	09876	PIC X (03)	エ
PIC 9 (05)	09876	PIC 9 (07)	オ
PIC 9 (05)	09876	PIC ZZZZ9	カ
PIC 9 (05)	09876	PIC ZZ9	キ
PIC 9 (05)	09876	PIC ZZ,ZZ9	ク
PIC 9 (05)	00000	PIC Z,ZZZ,ZZ9	ケ
PIC 9 (05)	00000	PIC Z,ZZZ,ZZZ	コ

2 次の代数式をCOMPUTE文で書きなさい。

(1) $A = B \times C$

(2) $A = B - C^2$

(3) $A = \frac{C+B}{D}$

(4) $A = B - \frac{C+D}{E \times F}$

(5) $A = \frac{B^2 - C^2}{D^3}$

(6) $A = \frac{B}{\frac{C}{D}}$

◀実習問題▶

1 健康診断ファイルを読んで、健康診断一覧表を作成しなさい。

▼入力形式

社員 コード	社員名	身長	体重
数字 5 桁	英数字10桁	数字 3 桁	数字 3 桁

▼処理条件

(1) ローレル指数

$$= \text{体重} * 10^7 / \text{身長}^3$$

(2) 身長, 体重, ローレル指数は編集して出力する。

▼出力形式

(社員コード)	(社員名)	(身長)	(体重)	(ローレル指数)
9 9 9 9 9	X X X X X X X X X X	Z Z 9	Z Z 9	Z Z 9
9 9 9 9 9	X X X X X X X X X X	Z Z 9	Z Z 9	Z Z 9
9 9 9 9 9	X X X X X X X X X X	Z Z 9	Z Z 9	Z Z 9
}	}	}	}	}

2 給与ファイルを読んで、給与一覧表を作成しなさい。

▼入力形式

職員 コード	名前	役職 コード	本給	加給	控除額
数字 4 桁	英数字 8 桁	英数字 1 桁	数字 7 桁	数字 6 桁	数字 6 桁

▼処理条件

(1) 支給額 = 本給 + 加給 - 控除額

(2) 本給, 加給, 控除額, 支給額は編集して出力する。

▼出力形式

(職員コード)	(名前)	(役職コード)	(本給)	(加給)	(控除額)	(支給額)
9 9 9 9	X X X X X X X X	X	Z, Z Z Z, Z Z 9	Z Z Z, Z Z 9	Z Z Z, Z Z 9	Z Z, Z Z Z, Z Z 9
9 9 9 9	X X X X X X X X	X	Z, Z Z Z, Z Z 9	Z Z Z, Z Z 9	Z Z Z, Z Z 9	Z Z, Z Z Z, Z Z 9
9 9 9 9	X X X X X X X X	X	Z, Z Z Z, Z Z 9	Z Z Z, Z Z 9	Z Z Z, Z Z 9	Z Z, Z Z Z, Z Z 9
}	}	}	}	}	}	}

2 集計

例題3 売上一覧表の作成

入力形式のようなデータ（売上ファイル）を読んで、売上一覧表を作成しなさい。

▼入力形式

商品コード	単 価	数 量
数字5桁	数字4桁	数字4桁

▼処理条件

- (1) 売上一覧表全体の見出しや各項目ごとの見出しをつける。
- (2) 金額は 金額＝単価×数量 で求める。
- (3) 出力結果のように、単価と金額には¥記号をつけ、単価では固定した場所に、金額では有効数字の直前に出力する。

また、数量は有効数字の前の0は空白におきかえ、3桁ごとにコンマを打つ。

▼出力形式

	0	1	2	3	4
	1234567890	1234567890	1234567890	1234567890	1234567890
1			*** ウリアケ	イチランヒヨウ ***	
2		コード	タンカ	スウリョウ	キンカク
3	00001	¥1,000	1,000	¥1,000,000	
4	00002	¥ 200	1,500	¥300,000	
5	00003	¥1,100	1,500	¥1,650,000	
6	}	}	}	}	
7	00057	¥ 700	750	¥525,000	
8	00058	¥1,850	200	¥370,000	
			コウケイ	¥53,946,250	

▼出力結果

	*** ウリアケ	イチランヒヨウ ***	
コード	タンカ	スウリョウ	キンカク
00001	¥1,000	1,000	¥1,000,000
00002	¥ 200	1,500	¥300,000
00003	¥1,100	1,500	¥1,650,000
}	}	}	}
00057	¥ 700	750	¥525,000
00058	¥1,850	200	¥370,000
		コウケイ	¥53,946,250


●問題の分析と考え方●

見出しや合計行の出力

通常、わかりやすい出力結果にするために、ページの先頭に一覧表全体のタイトルや、明細行の各項目のタイトルを印字する。また、一覧表の最後には、データ件数やデータの合計が印字される場合も多い。

この例題では、「*** ウリアケ イチランヒヨウ ***」という大見出し、「コードキンカク」という小見出しをページの先頭に印字し、最後に、「コウケイ」というタイトルとともに金額の合計を印字している。

これらは、明細行と同じように作業場所節に用意し、出力レコードを共通に用いてプリンタに印字される。



▲プログラムマップ

金額の集計

この例題では、最後に金額の合計を出力している。このような項目全体の集計を行うときは、プログラムマップに示したように、作業領域に集計するための記憶場所を設ける必要がある。

集計の手順は下記のとおりである。

- ① 「合計」(集計用の記憶場所) を0クリアする。

VALUE 句で0をセットしてもよい。

- ②から④をデータ終了まで繰り返す。

② データを読む。

③ 単価 * 数量 → 金額

④ 合計 + 金額 → 合計


- ⑤ 合計を印字する。

入力データが下記の場合、トレース表は右ようになる。


▼入力データ

単価	数量
100	10
20	15
110	5
}	}

単価	数量	金額	合計
			0
100	10		0
100	10	1000	0
100	10	1000	1000
20	15	1000	1000
20	15	300	1000
20	15	300	1300
110	5	300	1300



●流れ図●



●プログラム●

```

DATA
FILE
FD  URIAGE-FILE.
01  URIAGE-REC.
    10 SYOHIN      PIC 9(05).
    10 TANKA       PIC 9(04).
    10 SURYO       PIC 9(04).

FD  ITIRAN-FILE.
01  ITIRAN-REC     PIC X(132).
WORKING-STORAGE SECTION.
01  E-FLG          PIC X(03)  VALUE  "OFF".
01  KINGAKU        PIC 9(07).
01  KINKEI         PIC 9(08).
01  OOMIDASI-GYO   PIC X(39)  VALUE
    "          *** ウリアケ" イチランヒヨウ ***".
01  KOMIDASI-GYO   PIC X(49)  VALUE
    "          コート" タンカ          スウリヨウ          キンカ"ク".
01  MEISAI-GYO.
    10              PIC X(09)  VALUE  SPACE.
    10 M-SYOHIN     PIC 9(05).
    10              PIC X(04)  VALUE  SPACE.
    10 M-TANKA      PIC ¥Z,ZZ9.
    10              PIC X(05)  VALUE  SPACE.
    10 M-SURYO      PIC Z,ZZ9.
    10              PIC X(05)  VALUE  SPACE.
    10 M-KINGAKU    PIC ¥¥,¥¥¥,¥¥9.
01  GOKEI-GYO.
    10              PIC X(28)  VALUE  SPACE.
    10              PIC X(05)  VALUE  "コ"ウケイ".
    10              PIC X(05)  VALUE  SPACE.
    10 G-KINKEI     PIC ¥¥¥,¥¥¥,¥¥9.

```

*

PROCEDURE DIVISION. SYORI.

```

OPEN  INPUT  URIAGE-FILE  OUTPUT ITIRAN-FILE
MOVE  0      TO  KINKEI
WRITE ITIRAN-REC FROM OOMIDASI-GYO AFTER PAGE
WRITE ITIRAN-REC FROM KOMIDASI-GYO AFTER 1
PERFORM UNTIL E-FLG = "ON"
    READ  URIAGE-FILE
    AT END
        MOVE  "ON" TO E-FLG
    NOT AT END
        MOVE  SYOHIN TO M-SYOHIN
        MOVE  TANKA TO M-TANKA
        MOVE  SURYO TO M-SURYO
        COMPUTE KINGAKU = SURYO * TANKA
        MOVE  KINGAKU TO M-KINGAKU
        WRITE ITIRAN-REC FROM MEISAI-GYO AFTER 1
        COMPUTE KINKEI = KINKEI + KINGAKU
    END-READ
END-PERFORM
MOVE  KINKEI TO G-KINKEI
WRITE ITIRAN-REC FROM GOKEI-GYO AFTER 1
CLOSE URIAGE-FILE ITIRAN-FILE
STOP  RUN.

```

●プログラムの解説●

編集用文字「¥」

編集用文字「¥」は、通貨記号を指定した場所に出力する固定挿入編集と、有効数字の直前に出力する浮動挿入編集とがある。

「¥」を使った編集例を次に示す。

	送出し側		受取り側		備考
	PICTURE 句	記憶内容	PICTURE 句	転記の結果	
固定挿入	9 (04)	1234	¥ Z, Z Z 9	¥ 1, 234	左端に「¥」がつく
	9 (04)	0123	¥ Z, Z Z 9	¥ ΔΔ 123	左端に「¥」有効数字の前までの 0 は空白
	9 (04)	0000	¥ Z, Z Z 9	¥ ΔΔΔΔ 0	左端に「¥」、右端に「0」
浮動挿入	9 (04)	1234	¥ ¥ ¥, ¥ ¥ 9	Δ ¥ 1, 234	1 と 2 の間に「,」が入る
	9 (04)	0123	¥ ¥ ¥, ¥ ¥ 9	Δ Δ Δ ¥ 123	「,」の位置に「¥」が来る
	9 (04)	0001	¥ ¥ ¥, ¥ ¥ 9	Δ Δ Δ Δ Δ ¥ 1	¥ 1 でその前は空白
	9 (04)	0000	¥ ¥ ¥, ¥ ¥ 9	Δ Δ Δ Δ Δ ¥ 0	¥ 0 でその前は空白
	9 (04)	0000	¥ ¥ ¥, ¥ ¥ ¥	Δ Δ Δ Δ Δ Δ Δ	すべて空白に変わる
	9 (06)	123456	¥ ¥ ¥, ¥ ¥ 9	¥ 23, 456	先頭の 1 が切り捨てられる
	9 (06)	123456	¥, ¥ ¥ ¥, ¥ ¥ 9	Δ ¥ 123, 456	「¥,」ではじまる「,」の指定は意味がない

改ページ

WRITE 文(3)				
<u>WRITE</u>	出力レコード名	<u>FROM</u>	データ名	$\left\{ \begin{array}{l} \text{AFTER} \\ \text{BEFORE} \end{array} \right\} \text{PAGE}$

WRITE 文において、AFTER 句や BEFORE 句のあとに、PAGE 指定をすると、改ページ印字をする。AFTER 句の場合は改ページ後印字し、BEFORE 句の場合は印字後改ページをする。


●平均の計算●

集計処理の応用に、平均の計算処理がある。

平均を求めるには、データの合計のほかにデータの件数をカウントする必要がある。

「金額」の平均を求める手順は下記のようなになる。

- ① 「合計」と「件数」(データ件数) を 0 クリアする。
- ② から ④ までをデータ終了まで繰り返す。
 - ② データを入力する。
 - ③ 合計 + 金額 → 合計
 - ④ 件数 + 1 → 件数
- ⑤ 合計 / 件数 → 平均



●練習問題●

3 次の MOVE 文実行後の受取り側の内容はどうなっているか答えなさい。

送 出 し 側			受 取 り 側	
PICTURE 句		記憶内容	PICTURE 句	転記後の内容
PIC	9 (05)	98765	PIC ¥ZZ , ZZ 9	ア
PIC	9 (05)	98765	PIC ¥¥¥ , ¥¥ 9	イ
PIC	9 (05)	98765	PIC ¥ZZ 9	ウ
PIC	9 (05)	00765	PIC ¥¥¥ , ¥¥ 9	エ
PIC	9 (04)	0432	PIC ¥ZZ , ZZ 9	オ
PIC	9 (04)	0432	PIC ¥¥¥ , ¥¥ 9	カ
PIC	9 (04)	0000	PIC ¥ZZZ 9	キ
PIC	9 (04)	0000	PIC ¥¥¥¥ 9	ク

◀実習問題▶

3 健康診断ファイルを読んで、健康診断一覧表を作成しなさい。

▼入力形式

社員コード	社員名	身長	体重
数字 5 桁	英数字10桁	数字 3 桁	数字 3 桁

▼処理条件

- (1) ローレル指数
= 体重 * 10⁷ / 身長³
- (2) 見出しをつける。
- (3) 身長、体重、ローレル指数の全体の平均を求める。

▼出力形式

*** ケンコウシンタ'ン イチランヒヨウ ***				
コート'	シャインメイ	シンチヨウ	タイシ'ユウ	ローレルシスウ
99999	XXXXXXXXXX	ZZ9	ZZ9	ZZ9
99999	XXXXXXXXXX	ZZ9	ZZ9	ZZ9
99999	XXXXXXXXXX	ZZ9	ZZ9	ZZ9
}	}	}	}	}
	ヘイキン	ZZ9	ZZ9	ZZ9

4 ある出版社の 1 ヶ月出版物ファイルから出版物一覧表を作成しなさい。

▼入力形式

本コード	書籍名	定 価	発行日
数字 4 桁	英数字15桁	数字 5 桁	数字 2 桁

▼出力形式

*** シュツパ'ンブ'ツ イチランヒヨウ ***			
コート'	ショセキメイ	テイカ	ハツコウビ'
9999	XXX.....XXX	¥ZZ , ZZ9	99
9999	XXX.....XXX	¥ZZ , ZZ9	99
}	}	}	}
	ハツコウテンズウ		ZZ9
	ヘイキンテイカ	ZZ , ZZ9	

▼処理条件

- (1) 見出しをつける。
- (2) 定価は「¥」の固定挿入編集とする。
- (3) 最後に発行点数と、平均定価を出力する。